

Cominciano ad arrivare in massa i contributi dei lettori al software per TI-99/4A. Questo mese, vista anche la destinazione prettamente domestica ed hobbistica del Texas, abbiamo scelto due programmi che trasferiscono sul computer due dei giochi più noti e più classici: il domino e la tombola (Natale è ancora lontano, ma il tempo vola...).

Domino

Agostino Grilletta - Roma

Il DOMINO è un gioco per due persone. Lo scopo è quello di fare in modo che l'avversario non abbia più la possibilità di muoversi in caselle ancora libere, cercando di sbarrargli la strada o di chiuderlo in percorsi a vicolo cieco.



Il movimento delle pedine, rappresentate da un segnale a forma di rombo, è alterno e provoca la comparsa di una barriera nelle posizioni occupate precedentemente. Se avete visto il film Tron di Walt Disney potete farvi un'idea del suo principio, simile a quello del game delle motociclette, anche se decisamente più lento. Ovviamente non è possibile uscire dal campo di gioco poiché questo equivarrebbe ad andare a sbattere contro il muro. Inizialmente la direzione del movimento dei due segnali è determinata dal programma, per cambiarla occorre premere il tasto corrispondente; per spostare il segnale verso l'alto, il basso, a sinistra, a destra il giocatore a sinistra deve usare i tasti E, X, S, D rispettivamente; il giocatore a destra i tasti I, M, J, K. Per quanto riguarda il punteggio da raggiungere il programma chiede all'inizio di introdurre il numero relativo e quindi di assegnare una tot quantità di punti-premio per ogni manche; tale valore verrà inoltre incrementato di un numero di punti proporzionale alla durata della manche (uno per ogni mossa); il gioco termina quando dopo aver sommato le vincite dei vari round il punteggio ottenuto sia uguale o superiore a quello impostato inizialmente.

Passiamo ad esaminare, come di consueto, il listato, facendo riferimento alle singole linee di programma.

Analisi del listato

Nella variabile A\$ definiamo la forma del carattere usato per formare la barriera,


Domino

```


10 REM -DOMINO DI AGOSTINO GRILLETTA-
20 A$="FFFFC3C3C3C3FFFF"
30 B$="183C7EFFFF7E3C18"
40 CALL CHAR(104,A$)
50 CALL CHAR(105,B$)
60 CALL CHAR(96,A$)
70 CALL CHAR(97,B$)
80 CALL CLEAR
90 INPUT "PUNTEGGIO MASSIMO ? ":PT
100 INPUT "PUNTI-PREMIO MANCHE ? ":PP
110 CALL CLEAR
120 PRINT TAB(6);"SINISTRA DESTRA":;"ALTO";TAB(11);"E";TAB(18);"I"
130 PRINT :;"BASSO";TAB(11);"X";TAB(18);"M":;
140 PRINT :;"DESTRA";TAB(11);"D";TAB(18);"K"
150 PRINT :;"SINISTRA";TAB(11);"S";TAB(18);"J":;
160 PRINT "PREMI UN TASTO":;"PER COMINCIARE"
170 MAN=MAN+1
180 P=PP
190 CALL KEY(0,K,S)
200 IF S=0 THEN 190
210 CALL CLEAR
220 C(1)=6
230 C(2)=27
240 R(1)=4
250 R(2)=21
260 DIR(1,1)=0
270 DIR(1,2)=1
280 DIR(2,1)=0
290 DIR(2,2)=-1
300 CALL COLOR(1,11,11)
310 CALL COLOR(10,9,11)
320 CALL COLOR(9,6,11)
330 FOR I=1 TO 2
340 CALL KEY(I,K,S)
350 IF S=1 THEN 560
360 CALL HCHAR(R(I),C(I),112-B*I)
370 R(I)=R(I)+DIR(I,1)
380 C(I)=C(I)+DIR(I,2)
390 IF ABS(R(I)-12.5)>12 THEN 480
400 IF ABS(C(I)-16.5)>16 THEN 480
410 CALL GCHAR(R(I),C(I),CAS)
420 IF CAS<>32 THEN 480
430 CALL SOUND(100,200*I,5)
440 CALL HCHAR(R(I),C(I),113-I*8)
450 NEXT I
460 P=P+1
470 GOTO 330
480 CALL SOUND(900,-7,5)
490 G$="DESTRA"
500 IF I=1 THEN 520
510 G$="SINISTRA"
520 CALL COLOR(1,4,4)
530 CALL CLEAR
540 PRINT "IL GIOCATORE A ";G$;"GUADAGNA ";P;"PUNTI":;
550 GOTO 690
560 ON INT((K+3)/2)GOTO 570,600,630,660,360,360,360,360,360,360
570 DIR(1,1)=1
580 DIR(1,2)=0
590 GOTO 360
600 DIR(1,1)=0
610 DIR(1,2)=-1
620 GOTO 360
630 DIR(1,1)=0
640 DIR(1,2)=1
650 GOTO 360
660 DIR(1,1)=-1
670 DIR(1,2)=0
680 GOTO 360
690 PU(I)=PU(I)+P
700 PRINT "PUNTEGGIO ";MAN;"MANCHE":;TAB(11);"SINISTRA ";PU(2)
710 PRINT TAB(11);"DESTRA ";PU(1):;
720 IF PU(I)<PT THEN 160
730 PRINT "VINCE IL GIOCATORE A ";G$
740 FOR I=1 TO 4
750 CALL SOUND(600,200+100*I,5)
760 NEXT I
770 END
  
```





Lista di istruzioni per il movimento delle pedine



Fase del gioco



Altra fase del gioco



Assegnazione punteggi di una Manche

ossia un quadrato cavo, e quindi lo trasferiamo nei codici 96 e 104 del set ASCII; nella variabile B\$ definiamo invece lo Shape della pedina in movimento a forma di rombo che verrà assegnato ai caratteri 97 e 105 (linee da 20 a 70); il motivo che porta a definire due caratteri con lo stesso simbolo grafico consiste nel poter assegnare loro in tal modo due colori diversi, uno per ogni giocatore.

Dopo la richiesta di introduzione del punteggio da raggiungere nella variabile PT (90) e del punteggio da assegnare per ogni manche nella variabile PP (100), il programma provvede a visualizzare sullo schermo le istruzioni per guidare il movimento della pedina fino al momento in cui si preme un tasto qualsiasi (110-200). I vettori C ed R contengono rispettivamente la posizione orizzontale e verticale delle pedine dei due giocatori ed infatti sono costituiti da due soli elementi che nelle linee 220-250 vengono inizializzati con le coordinate di partenza (pregasi dare una sbirciata alla figura 1).

La matrice 2×2 di nome DIR contiene invece la direzione che il segnale deve assumere; l'elemento 1,1 rappresenta l'incremento della posizione verticale della pedina del giocatore a sinistra mentre quello della posizione orizzontale viene rappresentato dall'elemento 1,2. Stesso discorso vale per la pedina dell'avversario che, all'inizio della partita, si muove verso sinistra dal momento che il suo avanzamento in orizzontale (linea 290) assume valori negativi (pregasi dare nuovamente una sbirciata alla figura 1).

La riga 300 delimita il campo di gioco semplicemente assegnando al set di caratteri 1, e quindi al codice ASCII 32 con il quale viene rappresentato lo schermo vuoto,

il colore giallo scuro mentre nella 310 e nella 320 si stabilisce che il giocatore a sinistra sarà contraddistinto dal colore rosso, quello a destra dal blu chiaro. La tastiera, o meglio le semitastiere dal momento che la CALL KEY usa le unità 1 e 2 (340), vengono analizzate alternativamente e se non è rilevata la pressione di un tasto abilitato si procede all'avanzamento nella ultima direzione assegnata, controllando di non superare il bordo del campo da gioco (390-400) e di non andare ad occupare la posizione di un carattere che non sia quello dello spazio (420). Ad ogni mossa valida viene emesso un suono di frequenza variabile a seconda del segnale in avanzamento; se invece la mossa è errata viene generato un rumoraccio di collisione ed il controllo passa alla linea 500 che determina quale dei due contendenti ha perso la

manche con conseguente calcolo del punteggio e determinazione della necessità o meno (linea 720) di un nuovo round. Alla riga 560 troviamo la subroutine per il cambio di direzione, individuata in base al codice assegnato alla variabile di ritorno della CALL KEY della linea 340. Da notare che oltre ai tasti indicati nel quadro iniziale, sono anche attivi quelli assegnati alla A, W, R e alla H, U, O che corrispondono alla S, D, E della semitastiera sinistra e alla J, K, I della destra; attenzione quindi a non sbagliarvi.


Al termine della partita la riga 730 decreta la vittoria del giocatore a sinistra o a destra, mentre la 750 emette delle note di trionfo. Al perdente la possibilità di prendersi la rivincita lanciando nuovamente il RUN.

TOMBOLA

Antonio Bini - Firenze

Vogliamo confessarvi che non è stato facile prendere la decisione di pubblicare o meno questo mese il programma TOMBOLA di Antonio Bini. Parlare di tombola a giugno è un po' come parlare di windsurf a dicembre, ma dal momento che un buon programmatore deve essere un tipo previdente, alla fine abbiamo deciso che è pur sempre meglio prepararsi in anticipo; ci riserviamo però in tal caso il diritto di illustrarvi, per compensazione un algoritmo di tipo prettamente estivo sotto il periodo natalizio.

Il programma (come del resto anche quello di Agostino Grilletta) è un tipico esempio di come sia possibile ottenere con pochissime istruzioni (38) un risultato di



tutto rispetto ed inoltre indubbiamente interessante. Siamo sicuri che tutti coloro che possiedono un TI 99 non useranno più il classico tabellone con il sacchetti delle pedine numerate dal quale, inevitabilmente, il solito pargolo di turno ha sottratto proprio quella che ci mancava per fare tombola. Sono in tal modo evitate anche le

03	06 07			
12	16	18	20	
21	24 25	26 27	30	
32			39	
	44		48	
51 52 53 54 55	56			
	64		69	
	73 74		78 79	
82 83	85	86 87	89	
30				

Momento intermedio in cui è stato estratto il numero 30

01 02 03 04 05	06 07 08 09 10
11 12 13 14 15	16 17 18 19 20
21 22 23 24 25	26 27 28 29 30
31 32 33 34 35	36 37 38 39 40
41 42 43 44 45	46 47 48 49 50
51 52 53 54 55	56 57 58 59 60
61 62 63 64 65	66 67 68 69 70
71 72 73 74 75	76 77 78 79 80
81 82 83 84 85	86 87 88 89 90
68	

Tabellone al completo. Ultimo numero uscito: 68

"drittate" di colui che tiene il banco il quale non ha più la possibilità di operare movimenti strani che potrebbero far sorgere dubbi sulla... casuale casualità del caso.

Per concludere questa introduzione vogliamo esternarvi un dubbio che ci ha tormentati e che a volte ci ha impedito di dormire senza avere degli incubi terrificanti: sarà giusto automatizzare e togliere ogni spontaneità ad un gioco tipico e tradizionale come la tombola? Dopo aver sostituito le classiche lenticchie con le finestrelle di plastica ed il tabellone con lo schermo del TV, quanto tempo passerà prima che i premi in palio siano costituiti da fiches da spendere nella Arcade sotto casa?

Descrizione del programma

L'impiego del programma TOMBOLA è semplicissimo; dopo aver dato il RUN sullo schermo apparirà il tracciato del tabellone, suddiviso nelle sei tipiche cartelle. Ad ogni pressione di un tasto qualsiasi, verrà generato casualmente uno dei novanta numeri ancora da estrarre. Tale numero verrà mostrato in basso allo schermo, contemporaneamente sarà visualizzato in modo permanente nella sua esatta collocazione rispetto al tabellone (vedere foto) e una nota di durata e frequenza proporzionale al suo valore assoluto confermerà l'avvenuta estrazione. Al completamento del gioco occorre premere CLEAR

per fermare il programma e dare il RUN se si desidera farlo partire nuovamente. Da notare che non sempre l'estrazione del numero è contemporanea alla pressione del tasto poiché la sequenza di generazione casuale viene ripetuta fino a quando il valore ottenuto è tra quelli che ancora devono uscire il che, specie verso la fine della partita, non sempre avviene immediatamente. È perfettamente inutile pertanto mantenere premuto il tasto fino alla comparsa del numero estratto, tanto più che se non lo si rilascia in tempo esiste la possibilità di avere una estrazione doppia.

Analisi del listato

I caratteri ASCII 120 e 121 vengono definiti come un trattino orizzontale e uno verticale (linee 30 e 40), verranno poi impiegati per indicare la divisione fra le varie cartelle del tabellone. Notare come nella stringa esadecimale di definizione del codice 120 sia stato possibile omettere gli zeri non significativi alla destra di FF. Nella linea 50 si assegna il colore rosso scuro al set di caratteri numero 13; dal momento che sia il colore di background che quello di foreground sono uguali e che inoltre questo è un insieme di caratteri non definiti, avremo che con le righe 80-90 e 140 si otterrà una cornice rossa formata da tanti caratteri vuoti, ma caratterizzati da tale colore di sfondo.

Le altre due istruzioni CALL COLOR (60-70) assegnano ai caratteri definiti precedentemente e a quello dello spazio (ASCII = 32) la tonalità nera su sfondo grigio, creando in tal modo il tabellone e le linee di divisione (100-130).

Dopo aver così inizializzato lo schermo si passa a fare lo stesso con la matrice X che dovrà contenere i novanta numeri da estrarre, operazione che verrà effettuata non appena la CALL KEY della linea 200 permetterà il proseguimento del programma verso il loop di generazione casuale e controllo di validità del numero che, in caso di esito positivo, verrà sostituito con il valore zero (250) onde indicarne l'avvenuta estrazione. La riga 260 provvede ad emettere una nota di durata e frequenza proporzionale al numero generato mentre le rimanenti linee si occupano della sua visualizzazione nella posizione definitiva del tabellone e in quella provvisoria della casella di estrazione.

Tombola

```

10 REM -TOMBOLA DI ANTONIO BINI-
20 CALL CLEAR
30 CALL CHAR(120,"000000FF")
40 CALL CHAR(121,"1010101010101010")
50 CALL COLOR(13,7,7)
60 CALL COLOR(12,2,15)
70 CALL COLOR(1,2,15)
80 CALL HCHAR(24,1,132,64)
90 CALL VCHAR(1,32,132,48)
100 CALL HCHAR(20,2,120,30)
110 CALL HCHAR(8,2,120,30)
120 CALL HCHAR(14,2,120,30)
130 CALL VCHAR(2,17,121,19)
140 CALL HCHAR(22,1,132,32)
150 OPTION BASE 1
160 DIM X(90)
170 FOR I=1 TO 90
180 X(I)=I
190 NEXT I
200 CALL KEY(0,KEY,STATUS)
210 IF STATUS=0 THEN 200
220 RANDOMIZE
230 N=INT(RND*90)+1
240 IF X(N)<>N THEN 220
250 X(N)=0
260 CALL SOUND(N*10,N*10+100,1)
270 DECINE=INT(N/10)
280 UNITA=N-DECINE*10
290 RIGA=3+DECINE*2
300 COLONNA=UNITA*3
310 IF UNITA<>0 THEN 340
320 RIGA=RIGA-2
330 COLONNA=30
340 CALL HCHAR(RIGA,COLONNA,DECINE+48)
350 CALL HCHAR(RIGA,COLONNA+1,UNITA+48)
360 CALL HCHAR(23,16,DECINE+48)
370 CALL HCHAR(23,17,UNITA+48)
380 GOTO 200

```


*«Farò cose da meravigliare
il mondo intero...»*

...con il GPS-4 »



Michelangelo Buonarroti



GENERAL PROCESSOR

ELABORATORI ITALIANI